
HYPERPLANE ARRANGEMENTS

Questions

(1) A subset J = {j1, . . . , jp} ⊆ [n] := {1, . . . , n} is a circuit if (Hj1, . . . , Hjp
) is minimally

dependent. A subset J ⊆ [n] is a broken circuit if there exists jp+1 with ji < jp+1 for
1 ≤ i ≤ p and J ∪ {jp+1} is a circuit.
(a) Suppose that Q(A) = (x + y − z)xyz(x + y). Determine all the broken circuits

for A.
(b) Using Macaulay2, find a basis for the p-th graded component of the Orlik-

Solomon algebra
(

A(A)
)

p
for 0 ≤ p ≤ 3.

(c) What is the relationship between (a) and (b)? Can you prove it?

(2) Let A be an arrangement and let L = L(A). Define the Möbius function µ : L×L → Z

as follows:

µ(X, X) = 1 if X ∈ L,
∑

Z:X≤Z≤Y µ(X, Z) = 0 if X < Y ,

µ(X, Y ) = 0 otherwise.

For X ∈ L, define µ(X) = µ(V, X). The Poincaré polynomial of A is defined by
π(A, t) =

∑

x∈L

µ(X)(−t)rk(X).

(a) Define A by Q(A) = xyz(x − y)(x + y)(x − z)(x + z)(y − z)(y + z). Compute
µ(X) for all X ∈ L and π(A, t). Use Macaulay2 to compute the Hilbert series
of A(A).

(b) Define A by Q(A) = x1 · · ·x`. Show that for X ∈ L we have µ(X) = (−1)rk(X)

and π(A, t) = (1 + t)`.

(3) For X ∈ L(A) define a subarrangement AX of A by AX := {H ∈ A : X ⊆ H}. The
restriction of A to X is the arrangement AX in the vectors space X defined by

AX = {X ∩ H : H ∈ A \ AX} .

We call (A,A′,A′′) a deletion–restriction triple if A′ = A \ {H0} and A′′ = AH0 .
(a) If (A,A′,A′′) is a deletion–restriction triple, then show that

π(A, t) = π(A′, t) + tπ(A′′, t) .

(b) Let (A,A′,A′′) be a deletion–restriction triple. If A, A′, A′′ are the correspond-
ing Orlik-Solomon algebras and HS(A, t) :=

∑

p(dimC Ap)t
p, then prove that

HS(A, t) = HS(A′, t) + t HS(A′′, t).
(c) Prove that HS(A, t) = π(A, t).
(d) Define A and B by Q(A) = xyz(x − z)(x + z)(y − z)(y + z) and

Q(B) = xyz(x + y + z)(x + y − z)(x − y + z)(x − y − z) .

Compute π(A, t) and π(B, t).

(4) Consider the hyperplane arrangement A defined by

Q(A) = xyzw(x + y + z)(x + y + w)(x + z + w)(y + z + w) .

(a) Construct its Eisenbud-Popescu-Yuzvinsky module F (A) and compute its min-
imal free resolutions.

(b) Compute the Hilbert series of the Orlik-Solomon algebra A(A) and compare
with the betti number from part (a).

(c) Compute Ext2
S

(

F (A), S
)

. What can you say about the support (i.e. which
minimal primes lie over the annihilator)?

(d) Verify that the Orlik-Solomon ideal is generated in degree 3. What does this tell
you about the linear strand of Q as an A(A)-module?

(5) Suppose that Q(A) =
∏n

i=1 αi ∈ R = C[x1, . . . , x`]. The derivation module D(A)
is the R-module of all R-derivations θ such that θ(Q) ∈ (Q). The Euler derivation
∑`

i=1 xi∂/∂xi generates a free summand R(−1) of D(A) and D(A) = R(−1)⊕D0(A),
where D0(A) is the kernel of the transpose of the Jacobian matrix of Q. A hyperplane
arrangement A is free if and only D(A) is a free R-module.
(a) If Q(A) = xyz(x−y)(x+y)(x−z)(x+z)(y−z)(y +z) then is A free? Compute

and factor the Poincaré polynomial of A.
(b) Show that the braid arrangement is free.
(c) Write a Macaulay2 function that gives a matrix of generators for D(A) as a

submodule of Der(R). For the hyperplane arrangements in part (a) and (b),
compute the determinant of this matrix.

Macaulay 2 examples from the morning lecture

n = 4;

E = QQ[e_1..e_n, SkewCommutative => true];

e_1*e_2 + e_2*e_1 == 0

basis(1,E)

basis(2,E)

basis(3,E)

basis(4,E)

partial = m -> sum first entries compress diff(vars ring m, m)

partial (e_1*e_2)

partial (e_2*e_3*e_4)

partial (e_2*e_3*e_4 + e_1*e_2)

monomialSubIdeal = I -> (

R := ring I;

K := I;

J := ideal(1_R);

while (not isMonomialIdeal K) do (

J = ideal leadTerm gens gb K;

K = intersect(I,J));

ideal mingens K);

orlikSolomon = method(TypicalValue => Ideal);

orlikSolomon (List, Ring) := Ideal => (A,E) -> (

if (numgens E != #A) then error "incorrect number of variables";

C := substitute(syz jacobian matrix{A}, E);

M := monomialSubIdeal( ideal( (vars E) * C));

trim ideal apply(flatten entries gens M, r -> partial r));

orlikSolomon List := Ideal => A -> (

n := #A;

e := symbol e;

E := coefficientRing(ring A#0)[e_1..e_n, SkewCommutative => true];

orlikSolomon(A, E));

R = QQ[x,y,z];

A = {x+y-z,x,y,z,x+y}

I = orlikSolomon A

reduceHilbert hilbertSeries comodule I

basis(1, comodule I)

basis(2, comodule I)

basis(3, comodule I)

x = symbol x;

l = 4;

R = QQ[x_1..x_l];

A = toList flatten apply(1..(l-1),

i-> flatten toList apply(i+1..l, j-> {x_i-x_j}))

I = orlikSolomon A;

E = ring I;

transpose gens I

HS = reduceHilbert hilbertSeries comodule I

factor(numerator HS)

betti res (E^1/I)

J = ann(I);

transpose gens J

betti res J

S = QQ[e_1..e_#A];

sJ = substitute(ideal leadTerm J, S)

betti res sJ

sI = substitute(ideal leadTerm I, S)

betti res sI

sI == ideal dual monomialIdeal sJ

------------------------------------------------------------

partial = m -> (

E := ring m;

sum first entries compress diff(vars E,m));

monomialSubIdeal = I -> (

R := ring I;

K := I;

J := ideal(1_R);

while (not isMonomialIdeal K) do (

J = ideal leadTerm gens gb K;

K = intersect(I,J));

ideal mingens K);

orlikSolomon = method(TypicalValue => Ideal);

orlikSolomon (List,Ring) := Ideal => (A,E) -> (

if (numgens E != #A) then error "incorrect number of variables";

C := substitute(syz jacobian matrix{A},E);

M := monomialSubIdeal( ideal( (vars E) * C));

trim ideal apply(flatten entries gens M, r -> partial r));

orlikSolomon List := Ideal => A -> (

n := #A;

e := symbol e;

E := coefficientRing(ring A#0)[e_1..e_n,SkewCommutative=>true];

orlikSolomon(A,E));

R = QQ[x,y,z] -- coordinate ring

A3 = {x,y,z,x-y,x-z,y-z} -- A_3 braid arrangement

X3 = {x,y,z,y-z,x-z,2*x+y}

NF = {x,y,z,x-y,x-z,y-z,x+y-z} -- nonFano arrangement

HS = M -> numerator reduceHilbert hilbertSeries M;

I = orlikSolomon A3

E = ring I

A = E/I -- Orlik Solomon algebra

apply(1..numgens(A),i->e_i=first flatten entries (vars A)_{i-1})

H = w -> ann(w)/ideal(w);

HS H(e_1) -- not resonant

HS H(e_1-e_6) -- in R^2 but not R^1

HS H(e_1-e_2) -- in R^1; local component

HS H(e_1+e_6+e_2+e_5-2*(e_3+e_4)) -- in nonlocal component

-- F(A)

betti res module ann orlikSolomon A3 -- linear!

symExt= (m,R) ->(

if (not(isPolynomialRing(R))) then error "expected a polynomial ring or an exterior

if (numgens R != numgens ring m) then error "the given ring has a wrong number of variables";

ev := map(R,ring m,vars R);

mt := transpose jacobian m;

jn := gens kernel mt;

q := vars(ring m)**id_(target m);

n := ev(q*jn))

FA = method(TypicalValue => Module);

FA (Ideal,Ring) := Module => (j, R) -> (

modT := (ring j)^1/(j*(ring j^1));

F := res(prune modT, LengthLimit=>3);

g := transpose F.dd_2;

G := res(coker g,LengthLimit=>4);

coker symExt(G.dd_4, R));

FA (Ideal) := Module => (j) -> (

n := numgens ring j;

f := symbol f;

R := coefficientRing(ring j)[X_1..X_n];

FA(j, R));

fa = FA orlikSolomon A3

S = ring fa

betti (FF=res fa)

lc1 = coker matrix{{X_1+X_2+X_4,X_3,X_5,X_6}} -- S/p

apply({3,2,1,0},i-> hilbertPolynomial Tor_i(fa,lc1))

HS H (e_1-e_2) -- compare

-- this would be stupid:

FF.dd_1

-- decompose minors(6,FF.dd_1)

-- a more efficient way:

Hom(FF,S^1)

A3R1 = prune HH^2(oo)

boxList decompose ann A3R1

-- X3 example:

fa = FA orlikSolomon X3

betti (FF=res fa)

S=ring fa

X3R1 = prune Ext^2(fa,S)

boxList decompose ann X3R1 -- only local components

-- nonFano example:

fa = FA orlikSolomon NF

betti (FF=res fa)

S = ring fa

NFR1 = prune Ext^2(fa,S)

time ann NFR1;

